
第 1章

はじめに

この TJS活用講座は元々私のブログでひっそりと更新されていた TJS講座もどきなるものの延長線上に

あるものです。正直な話、“ひっそりと更新されていた”という部分は “殆ど更新されていなかった”にほぼ

等しいかもしれません。もうあれは忘れてください（＾＾；

対象としているのは主に KAGから TJSへ進んできた人です。一応 KAGをスキップして TJSを吉里吉

里でどうやって使うのか興味がある人も大丈夫だと思いますが。

さて、延長線上とは言いましたが TJS講座もどきで扱った内容も含んでいます。大幅に修正がされてい

ますが。流石にブログの内容をそのままそっくりコピペしてきたわけではありません。因みに今回こういう

体裁で文章を書いているのでいつもよりも真面目度が上がっています。表面的にはちょっと堅苦しく見える

かもしれませんが、前の段落で顔文字を使っていたように完璧真面目モードではありやせんぜ兄貴（ぁ

直前の妄言は忘れましょう。兎にも角にもブログの文章よりは真面目に書きます。

私事ですがこの文章は LATEXを使って書いています。私が LATEXの扱いに慣れてそこそこ熟練しようと

いう狙いも含みつつお送りします（笑

また、こういう体裁をとっているのでプリントアウトして読んでもかなり読みやすいはずです。ウィンド

ウの切り替えが煩わしい人等は紙媒体でどうぞ。

1.1 まえがき

ここでこうして教える側にいる私自身、プログラムのプの字も知らないところから TJSを学びました。参考書は吉

里吉里付属のリファレンスとネット上のいくつかのサイト、そして TJSを自由自在に使いこなしていた方々のサンプ

ルスクリプトでした。吉里吉里を使い始めたときは KAGも CUI *1ゆえにしばしば閉口していましたが、TJSはそれ

以上に難しくて挫折しかけました。不安を煽っているわけではありません。ただそれなりの努力を要するということを

言いたいのです。他にプログラム言語を齧ったことがあるなら大いに理解の助けとなります。しかし、これを読んでい

る人は十中八九そのアドバンテージを持っていないことと思います。そういった人達の理解の一助となれれば嬉しいで

す。話は逸れますが、他の人の書いたスクリプトはとても参考になります。余裕があればできるだけ参考にしてみて下

さい。

1.2 ここで TJSを学ぶ前に

ここで KAG の記述についての説明は基本的にしません。したがって、サンプルとして TJS の記述がされたから

といって ks ファイルにそのままコピペして動かないとぼやかないで下さい。ks ファイル内で TJS を記述するとき

[iscript]～[endscript]を使うことは常識として書いていきます。また TJSの文法については適宜リファレンス等

から引用します。ですから、文法のことを説明する章はありません。英語等の言語でもそうですが、文法はやっていて

*1 Character-based User Interface 要するにキーボードで主な操作をするユーザインターフェース

楽しくないですよね。大切ですけど。そういうわけで必要なときに必要なことを覚えるという方針でいきたいと考えて

います。

1.3 かまと

絵や図等はちょこちょこと入れていきます。文字ばかりだと嫌になってしまうでしょうから。そういう意味でキャ

ラクターを一人用意しました。余白で何かしていたり、文の合間に出現したりするかもしれません。名前はかまとで

す。*2かま子とでもお呼び下さい。

*2 キーボードのキーを T・J・Sの順にご覧下さい

第 2章

変数

すべての基礎、変数の話です。大抵プログラム等の講座では一番簡単なものとして最初に Hellow World

をやりますが、ここではやりません。文法的なこと等を Hellow Worldから学べますが、先に述べたように

文法の説明は必要に応じて挿入しますし、既に KAGを使っている状況で吉里吉里の起動方法等説明しても

釈迦に説法でしょう。変数と言って何のことだか全く分からない人もいないですよね。KAG でも f.～や

sf.～を使っていたことでしょう。もしかしたら mp.～を使っていた人もいるかもしれません。なので根本

的な変数の説明もスキップします。だたし、ここはどうしても文法的な事項に重きが置かれてしまいます。

最初だけ我慢してください。

2.1 TJSでの変数

TJSで扱う変数は KAGで使っていた変数とは少し違います。もちろん TJSで使っていた変数を KAGから使うこ

とはできますし、逆もまた然りです。そういう意味での違いではありません。KAGである変数を初めて使うときに変

数を宣言した覚えはありますか？普通の使い方をしていた人なら答えは NOのはずです。そもそも宣言って何さ？と

いう人もいておかしくありません。そのあたりから話を始めたいと思います。

本来変数は使いたいなと思ってすぐその場で使えるものではありません。まず使いたい変数を準備しておく必要があ

ります。KAGではそんな必要が無かったのにどうしてでしょう？その答えはこの先学ぶことになる配列で明らかにな

ります。現時点では準備をしなくてはならないものと割り切るしかありません。では、なぜ準備をする必要があるのか

という話になります。変数は PCのメモリ上のある場所に名前をつけてその場所に値を入れておく仕組みです。なので

変数を宣言してメモリのある場所を確保しなくてはなりません。

2.2 変数の宣言・初期化

では早速変数を宣言して使えるようにしてみましょう。

var elem;

この一行で elemという名前の変数を宣言しました。一行だけですが文法面で学ぶことがいくつかあります。

1. 一つの処理の最後にはセミコロン (;)が必要。

2. 意味が分からなくならない程度のホワイトスペース*1を入れることができる。

逆に、意味が分かるように適宜挿入する必要がある。

3. 変数の名前は予約語以外でなくてはならない。

ただし、数字は先頭にくることができない。

*1 改行やスペース、タブ等のこと

3について補足すると予約語とは以下のもの達のことです。

break continue const catch class case

debugger default delete do extends export

enum else function finally false for

global getter goto incontextof Infinity

invalidate instanceof isvalid import int in

if NaN null new octet protected property

private public return real synchronized switch

static setter string super typeof throw

this true try void var while with

これらの予約語はスクリプトの文法を構成する重要な用途として特殊な意味を持っています。ただし、予約語のほとん

どはがんばって覚える必要はありません。なぜなら、これらの予約語は TJSを学んでいけば特殊な意味を持った文字

列達だと分かるからです。例えば変数の宣言で使った varという文字列は

var ○○○;

のように書くことで○○○という名前の変数を宣言するという特別な意味を持ちます。このように予約語と呼ばれる文

字列は TJSで何か処理をさせる命令という風なものなのです。だからその内に覚えてしまっているというわけです。

因みに変数の名前に全角文字を使うこともできるにはできるのですが一般的ではありません。俺はマイナー道を突き

進むぜ！という人を無理に止めはしませんが、正直に言うと人に見られたとき非常に恥ずかしい思いをします。基本的

にコメント*2以外は半角文字を使いましょう。できれば英語の方が好ましいです。言い忘れるところでしたが、アル

ファベットの大文字と小文字は区別されます。

さて、冒頭の一行で変数は宣言されました。これで KAG でやっていたように変数を使った加減乗除等の演算がで

きるようになりました。この段階では変数 elem には何も値が入っていない状態です。言い換えると、この状態での

elemのデータ型は voidと言って何も表していないことを表しています。いきなりデータ型という単語を引っ張り出し

てきて混乱した人もいるかもしれませんね。変数は扱う値の種類によっていくつかの“型”があります。(整数型、実

数型、文字列型...etc) しかし、TJSでは変数の“型”が外部的にはありません。C言語等の他の言語では変数を宣言す

る際にそれぞれデータ型を指定してやる必要があるのですが、TJSでは指定する必要がないのです。しかし、TJSも

内部的には型を持っています。(よって型を持たないと言うよりは、型が自動的/動的に扱われると言った方が正確です

ね。これは TJSの長所の一つと言えると思います）その内部の型に void、整数型、実数型、文字列型、オブジェクト

型*3、オクテット型*4があります。そして宣言したての変数は voidです。

ただし、宣言と同時に何か値を変数に与えてやった場合はこの限りではありません。

var elem = 10;

これを初期化と言います。この場合、elemには既に 10という整数が入っています。ここで文法事項の 2について少

し説明しておきます。意味が分からなくならない程度のホワイトスペースとはどういうことでしょう？例として意味が

分からない程度にホワイトスペースを入れてみましょう。まず一つ目としては

var
elem
=
10
;

*2 //から行末までの部分、もしくは/*から*/で挟まれている部分。
このコメントの部分は TJSに解釈されないので自由にメモを残しておくといいです。
//についてはは KAGの;に似ていますね。

*3 ここでは気にしなくて OKです。参考までに。
*4 同じく参考までに書いたまでです。流して下さい。

というものです。読み辛いことこの上ありません。止めてください（＾＾；

v ar e le m = 1 0 ;

これが二つ目の例です。勿論これは論外です。正常な解釈ができません。つまるところ、こんなことにならないように

ホワイトスペースを入れましょうねってことです。また、意味が分かるように適宜挿入するというのはどいういうこと

でしょうか。再びやってはいけない例を示します。反面教師にして下さい。

varelem=10;

最早解説の必要はないでしょうが一応。varが変数を宣言するための特別な文字列です。しかしその後にスペースを設

けずに elemと入力してあります。これでは varと解釈されずに varelemとして解釈されます。意図した動作はして

くれません。

◆補足◆　数値と文字列

因みに記述する我々としては整数型と実数型はあまり意識する必要がありません。一番気にすることになるのは

整数型/実数型と文字列型です。これは KAGで経験済みかもしれませんが一応説明しておきます。数字の 1に

数字の 2を足すといくつになるでしょう。もちろん 3ですね。この 3は数字です。では、文字の 1に文字の 2

を足すといくつになるでしょう。答えは 12です。この 12は文字です。つまり、こういうことです。3を期待し

ていたのに結果として 12が得られてしまう、というような困ったことになりますね。こういう問題は必要に応

じて文字→数値の変換をしてやることで解決できます。単項演算子 “+”の出番です。加減算演算子の “+”と混

同すると危険かもしれません。加減算演算子は左側のものに右側のものを足す、という処理をします。要するに

小学生で習う足し算をするんですね。そして単項演算子は右側のものを文字列から数値に変換しようとがんばり

ます。
var elem = "10";
var num = +elem;

上の 2行ではまず elemに文字の 10が与えられ、numには elemを数値に変換して代入されています。文法的

なことで一つ確認しておくことが。それは文字列は"と"で囲むということです。KAGを普通に使っていれば学

習済みかと思いますが念のため。

2.3 ２重宣言

同じ名前の変数を２回以上宣言してもエラーが起こることはありません。ただし、宣言された名前の変数は voidと

なります。初期値を指定していればその値になります。スクリプトが大規模化すると無意識のうちに同じ名前の変数を

宣言してしまうことが起こりえます。そうするとエラーが出ないため予期せぬ挙動を示すことになりますね。気をつけ

ましょう。

2.4 変数のスコープ

まずスコープの説明の前にブロックというものについて説明の必要があります。ブロックとは{と}で囲まれた部分

のことをいいます。

var flag = 1;
if(flag == 1)
{

var elem;
}

上のスクリプトで言うと変数 elem が宣言されている部分です。ところで唐突に出現した if(flag == 1) とは何の

ことでしょう。お察しの通り条件分岐をしています。この==は KAGの ifタグで使っていた==と同じものです。TJS

ではこのように条件分岐を行います。条件式を評価して真なら後続ブロックを実行し、偽なら後続ブロックを無視し

ます。

ではスコープの説明に入っていきます。上の例では if の後続ブロック内で elem が宣言されています。実はこの

elemは ifの後続ブロックから出ると消えてしまいます。つまり、この elemはこのブロック内でのみ有効な変数*5な

のです。ところで、以下の場合を考えてみてください。

var flag = 1;
if(flag == 1)
{

var flag = 2;
}
var elem = flag + 1;

elemには何が入っていると思いますか？答えは 2です。えー (;´Д｀)と思った人はいますかね。恐らくいると思い

ます。何せ変数の 2重宣言を説明した直後ですからね。ブロック内で宣言された flagはローカル変数なのでブロック

の中でのみ有効です。なのでブロックの外で宣言された flagとは全く別の変数なのです。待てよ、と思った人はいま

すか？そうです。これではブロックの外で宣言した flagにアクセスできませんね。でもそんな不便な仕様にはなって

いません。一番外側で宣言された変数*6には global.～でブロックの中からでもアクセスできます。

var flag = 1;
if(flag == 1)
{

var flag = 2;
global.flag = flag;

}
var elem = flag + 1;

この場合には elemには 3が入ります。

◆補足◆　単項演算子 “++”

最後のスクリプトの最終行 var elem = flag + 1; は flag に 1 だけ足したものを elem に入れているため

var elem = ++flag; と書くことができます。単項演算子 “++” を使いました。便利な演算子なので是非使え

るようになりましょう。ただし、注意することがあります。足すものの前につけるか後ろにつけるかで意味が変

わってしまいます。例えば
var flag = 1;
var elem = flag++;

flag = 1;
var tmp = ++flag;

というスクリプトがあったとき、elemと tmpにはそれぞれ違う値が入っているのです。elemは 1で tmpは 2

です。それではいい加減 “++”について詳しく説明しましょう。そろそろイライラしてきたはずです（笑；

まずは flag++について。++が後置された場合、左にあるものに 1を加算して、全体としては加算する前の値

(左にあるものそのまま)になります。すなわち、elemには加算前の flagの値が入って、flagには 1が足され

るということです。

次は ++flagについてですね。++が前置された場合、右にあるものに 1を加算して、全体としては加算した後

の値になります。前置の方が分かりやすいですね。flagは 1が足されて、その 1足された flagの値が elemに

入るということになります。

蛇足かもしれませんが、“++”を使わずに上のスクリプトを書き換えてみます。
var flag = 1;
var elem = flag;
flag = flag + 1;

flag = 1;
flag = flag + 1;
var tmp = flag;

見比べてみて下さい。flag = flag + 1;の部分は “++”以外で書き換え可能ですが、それはまた別の機会に。

*5 これをローカル変数と言います。
*6 こちらはグローバル変数と言います。

第 3章

文字の表示

自分の書いたスクリプトで目に見える処理がされると嬉しくなります。特に意図した通りに動いたときに

は。前章で変数を通して TJSの基礎は学習しました。ここから先、吉里吉里に目に見える処理をしてもら

います。変数を宣言してもディスプレイの前の我々には何も変化していないように見えますね。当然のこと

ですがそれでは面白くありません。まずは前章の内容も踏まえて、変数の中身を吉里吉里に示してもらいま

しょう。

3.1 メッセージボックス

一番簡単な方法はメッセージボックスを使って変数の中身を表示することです。早速例を示します。

var text = "かまと";
System.inform(text);

吉里吉里で実行してみて下さい。“かまと”と書かれたメッセージボックスが開きましたね。2行目が謎なオーラを放っ

ていますがまだ分からなくていいんです。System.inform(～);で～の内容をメッセージボックスに表示する、とい

う覚え方で構いません。これをちゃんと理解するには関数やクラス/オブジェクトの知識が必要です。

とにかくこれで変数の内容が表示されてわかるようになりました。前章に戻って例として書いてあるスクリプトを試

してみるのもいいと思います。ちょっとだけ応用の例も書いておきます。

var text = "かまと";
var item = "バナナ";
System.inform(text + "は" + item + "を手に入れた。");

もし分かりにくかったら text + "は" + item + "を手に入れた。" の部分を変数で置き換えてみると見やすくなり

ます。

var text = "かまと";
var item = "バナナ";
var info = text + "は" + item + "を手に入れた。";
System.inform(info);

慣れないうちはこの方がいいかもしれませんね。

3.2 KAGの画面内に描画

吉里吉里の画面と言ったほうがよくわかるかもしれませんが、一応厳密に KAGの画面と書きました。ただの表現の

違いというくらい微妙なことだと思うのであまり深く考えません（笑

3.2.1 メッセージレイヤへの描画

勿論 KAG でできることなのでわざわざ TJS を使う必要はありません。ただ、TJS で書いている途中に（例えば

KAGプラグイン等）KAGは使いません。できないことではありませんが、普通はそのままの流れで TJSで処理を書

きます。では先程のかま子がバナナを入手したことについてメッセージレイヤに描画してみましょう。

var text = "かまと";
var item = "バナナ";
var info = text + "は" + item + "を手に入れた。";
kag.fore.messages[0].drawText(0, 0, info, 0xffffff);
kag.fore.messages[0].visible = true;

当然ながら変数を宣言するところまでは前のメッセージボックスに表示したときと同じです。4行目では、表画面メッ

セージレイヤ 0の x=0、y=0の位置に infoの内容を白文字で描画しており、次の行で可視状態にしています。折角

描画しても不可視状態のままでは見えませんからね。kag.fore.messages[0]の部分を見て、KAGに慣れ親しんで

いる皆さんはどのレイヤに描画しようとしているのか、については最早説明を要しないことと思います。裏画面のメッ

セージレイヤ 1に描画したければ

var text = "かまと";
var item = "バナナ";
var info = text + "は" + item + "を手に入れた。";
kag.back.messages[1].drawText(0, 0, info, 0xffffff);
kag.back.messages[1].visible = true;

としてやればいいのです。[endscript] の後でトランジションでもやれば確認できることでしょう。ところで、

drawTextの後ろのカッコの中身は一体なんでしょう。先述の説明で何となく勘付いている人もいることと思いますが

説明します。コンマ “,”で区切られているので便宜的に前から a,b,c,dとします。aは文字を描画する x座標、bは描

画する y座標、cは描画する文字列、dが描画する色となっています。なので infoの内容を表画面メッセージレイヤ

1の x=40、y=100の位置に赤文字で描画したければ

var text = "かまと";
var item = "バナナ";
var info = text + "は" + item + "を手に入れた。";
kag.back.messages[1].drawText(40, 100, info, 0xff0000);
kag.back.messages[1].visible = true;

という具合になります。

◆補足◆　少し掘り下げた drawTextの使い方

吉里吉里２リファレンスでクラスリファレンスの Layerクラスからメソッド drawText(文字描画)という項目の

内容を見てみるとアレ？と思うかもしれません。実際にご覧下さい。構文のところに書かれているものが妙に長

たらしいですね。ここで使ったものとは偶然名前が同じだけで別物…というわけではありません。実は今回使っ

た drawText は color 以降の部分を省略したのです。ですから最後の shadowsfy のところまで値を書いても

OKです。ただし、注意があります。opa=255や aa=true等と書いてあるからといって

kag.fore.messages[0].drawText
(

0,
10,
"てすと",
0xffffff,
opa=200,
aa=true,
shadowlevel=100,
shadowcolor=0x505050,
shadowwidth=2,
shadowofsx=2,
shadowofsy=4

);

と書かないようにして下さい。(長くなってしまい 1行に収まらないため、カッコの中を 1つのブロックのよう

に見える書き方をしました。このようなブロックのように見える書き方をするな、と言っているわけではありま

せん。勘違いしないように。) TJSは KAGと違ってマークアップ言語*1ではありません。正しくは

kag.fore.messages[0].drawText
(

0,
10,
"てすと",
0xffffff,
200,
true,
100,
0x505050,
2,
2,
4

);

という風になります。では、何故 opa=255のような表記がされているのでしょう。それは省略した場合赤字で

書いてある値を指定したことになる、ということを意味しています。そして、値のみをコンマで区切って書いて

いくため順番が大切になります。因みにここで出てきたクラス・メソッドという言葉はまだ理解していなくてい

いのであまり気にしないで下さい。この章で覚えるべきは “どうすれば TJSを使って文字の描画 (表示)ができ

るのか”です。大量のことを一気に覚えさせようとはしませんので整理しながら進んでいって下さい。でもとき

には振り返ることも必要です（ぁ

3.2.2 その他レイヤへの描画

まずはメッセージレイヤと同様に前景レイヤに文字を描画しましょう。実はこの時点でできてしまう人もいるかもし

れません。例えば表画面の前景レイヤ 0にかま子がバナナを何たらという文を表示する場合

var text = "かまと";
var item = "バナナ";
var info = text + "は" + item + "を手に入れた。";
kag.fore.layers[0].drawText(40, 100, info, 0xff0000);
kag.fore.layers[0].visible = true;

という感じになります。そうです、messagesが layersに代わっただけです。KAGの名残で messagesを message、

layersを layerと書かないように気をつけて下さい。そして、表画面の背景レイヤに同様に表示するなら

var text = "かまと";
var item = "バナナ";
var info = text + "は" + item + "を手に入れた。";
kag.fore.base.drawText(40, 100, info, 0xff0000);
kag.fore.base.visible = true;

非常に簡単ですがこれで終わりです (汗

あらかたのことを先に説明してしまいましたのでこの章はこれで終了です。じゃー次行ってみよー。

*1 KAGではタグを使うとき、属性とそれぞれの値を書いていましたがこのような記述をするのはマークアップ言語の場合のみです。

第 4章

画像の表示

言うまでもなく文字が表示されただけではどうしようもありません。勿論文字さえ表示することができれ

ば画像の代わりに文字を使ってアイディア勝負のゲームを作ることができると言えばできます。ただやっぱ

り普通にゲームを作るなら画像の読み込みは避けては通れない道です。そこでここでは画像をレイヤに読み

込む処理から、できると便利な処理を少しだけ紹介します。あ、言うまでもないと思いますけど何か画像を

用意して下さいよ（＾＾；

4.1 レイヤへの読み込み

用意した画像を対象のレイヤに読み込ませましょう。しかし、すでにある KAGのレイヤ*1に TJSで記述して読み

込ませるのはちょっと面倒です。なので自分でレイヤを用意して、そのレイヤに画像を読み込ませましょう。こっちは

簡潔に記述することができます。レイヤを用意するには本来クラスやオブジェクトといった知識が必要なわけですがこ

こでは詳しく説明しません。後々クラスについて触れる際にわかることでしょう。では、レイヤを作ってみます。

var lay = new Layer(kag, kag.fore.base);

完了しました。今のところは “おなじないの一行”で結構です。layという名前のレイヤを使うことができます。とこ

ろで、varが出てきていますね。しかしできたのは変数ではなくレイヤだと言っています、この人は。なんのこっちゃ

わけわからんわ、となっていますか？先に触れたオブジェクトという単語から変数のデータ型の話で出てきた言葉 “オ

ブジェクト型”を思い浮かべた方は勘付いているかもしれませんが、ここで解説することはしません。おまじないなの

です。それより画像を読み込みましょう。

var lay = new Layer(kag, kag.fore.base);
lay.loadImages("sample.png");
lay.setSizeToImageSize();
lay.visible = true;

一行ずつ見ていきます。実際に画像を読み込む処理をしているのは 2 行目だけです。ではそれ以降は何をしているの

でしょう。ここは流石におなじないで済ませるわけにはいきません。4行目は説明無しでも大丈夫かもしれませんね。

作ったレイヤ layを可視状態にしています。折角画像を読み込んでもレイヤが不可視では見えませんからね。3行目は

読んだままですが説明をば。レイヤのサイズを画像のサイズと同じにしています。何故こんなことをしなくてはならな

いのか。

画像サイズがレイヤの表示サイズより小さかった場合を 除いて、レイヤの表示サイズは変更しません。（吉里吉

里 2 リファレンスより）

*1 例えば前景レイヤ等

要するにレイヤのサイズが小さいのにそれより大きい画像を読み込むとレイヤのサイズ分しか表示されないため、で

す。一度 3行目をコメントアウト*2して実行してみると違いがわかると思います。そういうことなので、基本的には 2

行目と 3行目はほぼセットで記述します。

◆補足◆　 withステートメント

実は 2行目から 4行目については省略した書き方ができます。それには withを使います。別にあのままでいい

んですが、記述するときに何度も layと打つのは面倒じゃありませんか？これがたった３字なのでそれほど面

倒さを感じないかもしれませんが表画面の前景レイヤ 0だったとき等長い場合は断言します、面倒だと。そして

修正するときも面倒で仕方ありません。というわけで、これができなくてもゲームに影響はありませんが効率が

変わってくるわけです。では、先にスクリプトを示します。

var lay = new Layer(kag, kag.fore.base);
with(lay)
{

.loadImages("sample.png");

.setSizeToImageSize();

.visible = true;
}

こんな感じに記述することができます。見た目としてはピリオド（.）の前を省略して書くことができそうです

ね。仕組みはオブジェクトについて説明する際に一緒に説明することにしましょう。ここでは見た目のままの使

い方で構いません。

さて、KAGのレイヤに TJSで画像を読み込ませるのは面倒だと言いました。実際のところ記述自体にはそれほど面

倒なことはありません。配列の内容を先取りした内容になってしまいますのでここでは割愛しました。

4.2 レイヤへの操作

レイヤは画像を読み込んだり、文字を表示するだけのものではありません。これは KAGを弄っていた人には既にわ

かっていることでしょう。TJSを使うことで KAG標準ではできないレイヤ操作をすることが可能です。ここではそ

の一部を取り扱います。基本的なやり方がわかれば、あとは全部その応用です。

4.2.1 塗りつぶし

まずは肩慣らしです。表画面の背景レイヤの一部を真っ白に塗りつぶしてみます。

kag.fore.base.fillRect( 50, 0, 300, 450, 0xffffffff );

これで完了です。実際に実行してみると各引数*3が何を意味しているかわかると思うのであえてその説明はしません。

あ、でも最後のものは見慣れない代物かもしれませんね。一応 0xffffffffだけ説明をしておきます。0xffffffの間

違いなんじゃないか？と思った方もいるでしょうが、間違いではありません。*40xffffffは 0xRRGGBBという色の指

定に使っていました。一方 0xffffffffは 0xAARRGGBBという形式をとっています。AAとは何のことか…これは透

明度のことです。ffが完全不透明、00が完全透明。もう大丈夫ですね。では表画面の背景レイヤ全部を真っ白に塗る

のにはどうすればいいでしょう。

kag.fore.base.fillRect( 0 , 0, 640, 480, 0xffffffff );

ですね。でもちょっと待って下さい。何気なく 640や 480という数値を直に記述しましたが、実はあまりいい書き方

ではありません。もし画面サイズを変更した場合、当然レイヤのサイズも変わってきます。何せ背景レイヤですから

ね。その都度手打ちで修正するのは骨が折れます。なのでレイヤ全体を塗りつぶすのであればレイヤの大きさを取得し

*2 3行目の頭に//を記述するなどしてコメント行にしてやること
*3 カッコの中の,で区切られたそれぞれの数値や文字列のことです。
*4 それまでの記述如何では 0xffffffと書く方が正しい場合もあります。今の時点では気にしなくていいのですが気になるのであれば吉里吉里
２リファレンスの Layerクラス―fillRect(矩形塗りつぶし)を参照して下さい。

てそれを利用しましょう。

var wid = kag.fore.base.width;
var hei = kag.fore.base.height;
kag.fore.base.fillRect( 0 , 0, wid, hei, 0xffffffff );

見てもらった通りです。kag.fore.base.widthで横幅を、kag.fore.base.heightで縦幅を取得します。サンプル

ではそれぞれを変数に入れて、各変数を利用しています。別に変数に入れる必要もないので次のように書くこともでき

ます。

with(kag.fore.base)
{

.fillRect( 0 , 0, .width, .height, 0xffffffff );
}

この場合中カッコは無くてもいいんですけど、可読性が良くなるので私は滅多に省略していません。邪魔だから無くて

いいものは書かなくていいじゃないか、と思うのであれば次のように書くとスッキリします。

with(kag.fore.base) .fillRect( 0 , 0, .width, height, 0xffffffff );

4.2.2 反転

予定

4.2.3 ブラー

予定


